

# Newsroom druckt deutsch

Einer der Hauptmängel des Zeitungsprogramms »Newsroom« wird mit zwei Listings behoben: Ab sofort kann man auch mit Umlauten arbeiten.

Da hat man nun ein tolles Druckprogramm, mit dem man selber Zeitungen drucken kann, und dann beherrscht es keine Umlaute. So ist die Situation beim »Newsroom«, den wir in Ausgabe 3/86 ausführlich vorstellten. Doch mit etwas Geduld und den zwei hier abgedruckten Listings kann man dem Newsroom zu den Umlauten, aber auch zu völlig neuen Zeichensätzen verhelfen.

Für die zwei Größen von Zeichensätzen sind zwei verschiedene Listings notwendig. Mit »Small Change« (Listing 1) können Sie die kleinen, mit »Big Change« (Listing 2) die großen Zeichensätze editieren.

Die Diskette mit den Zeichensätzen muß sich in Laufwerk 8 befinden. Dazu sollte man die Zeichensatzfiles »SFNTS«, »LFNT0«, »LFNT1« und »LFNT2« mit einem File-Kopierprogramm von der Original-Diskette auf eine andere, leere Diskette überspielen. Die beiden kleinen Zeichensätze sind im File »SFNTS« enthalten. Das Programm »Big Change« fragt nach der Zeichensatznummer zwischen 0 und 2, um dann das entsprechende Zeichensatzfile zu laden.

Das Lesen der Zeichen von Diskette dauert ein Weilchen, bedingt durch das langsame Basic-Programm. Danach erscheint das erste Zeichen (»!«) auf dem Bildschirm. Ab jetzt kann man das Programm wie folgt über die Tastatur steuern: <Space> läßt das nächste Zeichen auf dem Bildschirm erscheinen.

<\*> dient zum direkten Anspringen eines Zeichens. Dazu muß die Zeichennummer angegeben werden (entspricht nicht den ASCII-Codes!). Die kleinen Zeichensätze haben 200 Zeichen, die großen 100. Größere Zahlen können zum Programmabsturz führen.

**Die Cursorstasten** dienen zur Steuerung des Cursors im umrahmten Feld.

<=> setzt oder löscht den Punkt an der Cursorposition.

<Return> übernimmt das editierte Zeichen wieder in den Speicher.

Soll das editierte Zeichen unter einer anderen Zeichennummer gespeichert werden, betätigen Sie den Klammerschalter und geben die Nummer ein. Diese Funktion ist besonders nützlich, um die Umlaute zu erstellen.

<1> speichert den editierten Zeichensatz unter dem Namen »NEWFONT« (und einem Anhängsel) auf Diskette. Danach müssen Sie ihm mit dem Rename-Befehl der Floppy den Namen des editierten Zeichensatzes geben.

Leider hat der Newsroom eine eigenwillige Tastaturabfrage, so daß sich die Umlaute nur schwer auf der Tastatur unterbringen lassen. Deswegen hier ein Vorschlag für Zeichen, die Sie für die Umlaute opfern sollten:

Taste	Zeichen	Nummern
Klammeraffe	ß	35,131
Pfeil n. oben	ä	65,161
SHFT Pfeil n.o.	Ä	97,193
Pfund	ü	63,159
SHFT Pfund	Ü	95,191
eck. Klammer l.	ö	62,158
eck. Klammer r.	Ö	64,160

Als Vorlagen für die Umlaute kann man die normalen Vokale verwenden. Sie haben die folgenden Nummern: a = 68, A = 36, o = 82, O = 50, u = 88, U = 56

Um also ein »ä« zu erstellen, gehen Sie wie folgt vor: Wählen Sie Zeichen 68 (»a«) an, setzen Sie Punkte darauf und speichern es unter der Nummer 65.

Sollte Ihnen das alles viel zu viel Arbeit sein: Auf der Leser-service-Diskette befinden sich die fertigen deutschen Zeichensätze, die nach oben genannter Tabelle erstellt wurden.

Das letzte Problem, das sich Ihnen noch stellt: Wie manipulieren Sie Ihre Newsroom-Diskette, daß die deutschen Zeichensätze auch wirklich benutzt werden? Sie sollten auf gar keinen Fall Ihre Original-Diskette verwenden, da Sie diese sonst zerstören könnten. Machen Sie also mit einem Kopierprogramm eine Kopie der Diskette. Stören Sie sich nicht an Fehlermeldungen auf Spur 35, die gehören nämlich zum Kopierschutz. Die Kopie ist zwar nicht alleine lauffähig, aber Sie können die Zeichensatzdaten dort ändern. Löschen Sie also die alten Zeichensätze und speichern Sie die neuen mit einem File-Kopierprogramm.

Wenn Sie mit deutschen Zeichensätzen arbeiten wollen, gehen Sie wie folgt vor: Laden Sie den Newsroom normal ein; sobald das Titelbild mit der Menüauswahl erscheint, tauschen Sie die Original-Diskette gegen Ihre Kopie mit den neuen Zeichensätzen aus. Nun sollten Sie problemlos arbeiten können.

Da es beim Original-Newsroom inzwischen verschiedene Versionen gibt, können wir nicht garantieren, daß dieses Verfahren immer zum Erfolg führt. Sollten uns Unverträglichkeiten bekannt werden, werden wir versuchen, diese in einem Folge-Artikel zu beseitigen. Eines ist auf jeden Fall sicherzustellen: Manipulieren Sie nur an einer Kopie, niemals an der Original-Diskette! Sollten Sie diese zerstören, kann es einige Zeit dauern und auch ein paar Mark kosten, bis Sie vom Händler eine Ersatz-Diskette erhalten. (bs)

```

10 REM *** ZEICHENSATZAENDERUNG *** <158>
20 REM *** FUER "THE NEWSROOM" *** <177>
30 REM *** KLEINE SCHRIFT *** <252>
50 POKE 53280,3:POKE 53281,3:PRINT "CLR,BL
UE)" <007>
100 DIM B(2000) <254>
110 OPEN 3,8,3,"SFNTS,P,R" <059>
115 PRINT "{3DOWN}BITTE GEDULD - PURES BASI
C" <059>
120 AN=AN+1:GOSUB 1000:B(AN)=A <001>
130 IF ST=0 THEN 120 <025>
140 CLOSE 3 <167>
200 FOR L=4 TO 199 <077>
210 IF B(L)=0 THEN 310 <151>
220 GOSUB 2000:PRINT "{HOME,9DOWN}" <166>
230 FOR K=0 TO 7 <059>
240 FOR M=0 TO 7 <085>
250 IF (B(L*8+171+K)AND 2^M)=0 THEN 270 <049>
260 POKE 1080+K*40+(7-M),81 <193>
270 POKE 55352+K*40+(7-M),6 <115>
300 NEXT M,K <205>
305 PRINT "ZEICHEN #:"L <008>
310 PRINT "{DOWN}BREITE {3SPACE}:"B(L) <216>
315 X=0:Y=0:POKE 1080,PEEK(1080)OR 128 <218>
320 GET A$:IF A$=""THEN 320 <131>
325 IF A$=""THEN INPUT "{2DOWN}ZEICHEN #:"
;L:GOTO 210 <020>
326 IF A$="@ THEN INPUT "{2DOWN}NEUE ZEICHE
N #:";L:GOSUB 1700:GOTO 410 <136>
330 IF A$="↑ THEN L=199:GOTO 410 <229>
340 IF A$="<RIGHT>" THEN GOSUB 1100 <145>
350 IF A$="<LEFT>" THEN GOSUB 1200 <235>
360 IF A$="<DOWN>" THEN GOSUB 1300 <191>
370 IF A$="<UP>" THEN GOSUB 1400 <025>
380 IF A$="=" THEN GOSUB 1600 <025>
390 IF A$=CHR$(13) THEN GOSUB 1700:GOTO 410 <060>
395 IF A$=" " THEN 410 <074>
400 GOTO 320 <122>
410 NEXT L <008>
420 OPEN 3,8,3,"@:NEWFONT,P,W" <106>
430 FOR L=1 TO AN <128>
440 PRINT#3,CHR$(B(L)); <250>
450 NEXT <206>
460 CLOSE 3 <233>
470 END <218>
880 NEXT L:END <149>
1000 GET#3,A$ <167>

```

```

1010 IF A$="" THEN A=0:RETURN <073>
1020 A=ASC(A$):RETURN <070>
1100 REM *** CURSOR RECHTS *** <097>
1110 GOSUB 1500 <130>
1120 IF X<6 THEN X=X+1 <003>
1130 GOSUB 1500 <150>
1140 RETURN <182>
1200 REM *** CURSOR LINKS *** <052>
1210 GOSUB 1500 <230>
1220 IF X>0 THEN X=X-1 <139>
1230 GOSUB 1500 <250>
1240 RETURN <026>
1300 REM *** CURSOR RUNTER *** <229>
1310 GOSUB 1500 <076>
1320 IF Y<7 THEN Y=Y+1 <119>
1330 GOSUB 1500 <096>
1340 RETURN <128>
1400 REM *** CURSOR RAUF *** <036>
1410 GOSUB 1500 <176>
1420 IF Y>0 THEN Y=Y-1 <223>
1430 GOSUB 1500 <196>
1440 RETURN <228>
1500 REM *** CURSOR SETZEN/LOESCHEN *** <015>
1510 IF PEEK(1080+X+Y*40)>128 THEN POKE 10
      80+X+Y*40,PEEK(1080+X+Y*40)AND 127:RE
      TURN <140>

```

```

1520 POKE 1080+X+Y*40,PEEK(1080+X+Y*40)OR
      128:RETURN <060>
1600 REM *** PUNKT SETZEN/LOESCHEN *** <234>
1610 IF PEEK(1080+X+Y*40)=160 THEN POKE 10
      80+X+Y*40,209:RETURN <122>
1620 IF PEEK(1080+X+Y*40)=209 THEN POKE 10
      80+X+Y*40,160:RETURN <031>
1700 REM *** NEUES ZEICHEN *** <070>
1710 GOSUB 1500:INPUT "{HOME,12DOWN}BREITE {
      3SPACE}";B(L) <188>
1720 FOR K=0 TO 7:B(L*8+171+K)=0 <125>
1730 FOR M=0 TO 7 <051>
1740 IF PEEK(1080+K*40+(7-M))=81 THEN B(L*
      8+171+K)=B(L*8+171+K)OR 2↑M <004>
1750 NEXT M,K <129>
1760 RETURN <038>
2000 REM *** BILDSCHIRMAUFBAU *** <071>
2010 PRINT "{CLR,15SPACE,RVSON,9SPACE}" <098>
2020 FOR K=1 TO 8 <232>
2030 PRINT "{15SPACE,RVSON,SPACE,7RIGHT,SPA
      CE}" <225>
2040 NEXT <016>
2050 PRINT "{15SPACE,RVSON,9SPACE}" <238>
2060 RETURN <086>

```

Listing 1. Änderungsprogramm für die kleinen Zeichensätze

```

10 REM *** ZEICHENSATZAENDERUNG *** <158>
20 REM *** FUER "THE NEWSROOM" *** <177>
30 REM *** GROSSE SCHRIFT *** <048>
50 DIM B(3500) <045>
60 POKE 53280,3:POKE 53281,3 <206>
70 PRINT "{CLR,BLUE}" <097>
90 INPUT "{CLR,2DOWN}ZEICHENSATZ (0-2):";S$ <178>
100 F$="LFNT"+S$+" ,P,R" <186>
110 OPEN 3,8,3,F$ <058>
115 PRINT "{2DOWN}BITTE GEDULD - IST HALT N
      UR BASIC" <054>
120 AN=AN+1:GOSUB 1000:B(AN)=A <001>
130 IF ST=0 THEN 120 <025>
140 CLOSE 3 <167>
200 FOR L=4 TO 132 <057>
210 IF B(L)=0 THEN 310 <151>
220 GOSUB 5000:PRINT "{HOME,18DOWN}" <239>
230 V=0:W=0:GOSUB 2000 <204>
240 V=8:W=1:GOSUB 2000 <087>
250 V=320:W=16:GOSUB 2000 <067>
260 V=328:W=17:GOSUB 2000 <089>
305 PRINT "ZEICHEN #:"L <008>
310 PRINT "{DOWN}BREITE {3SPACE}:"B(L) <216>
315 X=0:Y=0:POKE 1076,PEEK(1076)OR 128 <234>
320 GET A$:IF A$="" THEN 320 <131>
325 IF A$="" THEN INPUT "{DOWN}ZEICHEN #:";
      L:GOTO 210 <037>
326 IF A$="@ THEN INPUT "{DOWN}NEUE ZEICHEN
      #:";L:GOSUB 1700:GOTO 410 <226>
330 IF A$="↑ THEN L=132:GOTO 410 <098>
340 IF A$="> THEN GOSUB 1100 <145>
350 IF A$="< THEN GOSUB 1200 <235>
360 IF A$="↓ THEN GOSUB 1300 <191>
370 IF A$="↑ THEN GOSUB 1400 <025>
380 IF A$="↓ THEN GOSUB 1600 <025>
390 IF A$=CHR$(13) THEN GOSUB 1700:GOTO 410 <060>
395 IF A$=" " THEN 410 <074>
400 GOTO 320 <122>
410 NEXT L <008>
415 F$="@:NEWFONT"+S$+" ,P,W" <098>
420 OPEN 3,8,3,F$ <114>
430 FOR L=1 TO AN <128>
440 PRINT#3,CHR$(B(L)); <250>
450 NEXT <206>
460 CLOSE 3 <233>
470 END <218>
1000 GET#3,A$ <167>
1010 IF A$="" THEN A=0:RETURN <073>
1020 A=ASC(A$):RETURN <070>
1100 REM *** CURSOR RECHTS *** <097>
1110 GOSUB 1500 <130>
1120 IF X<15 THEN X=X+1 <095>
1130 GOSUB 1500 <150>
1140 RETURN <182>
1200 REM *** CURSOR LINKS *** <052>
1210 GOSUB 1500 <230>
1220 IF X>0 THEN X=X-1 <139>

```

```

1230 GOSUB 1500 <250>
1240 RETURN <026>
1300 REM *** CURSOR RUNTER *** <229>
1310 GOSUB 1500 <076>
1320 IF Y<15 THEN Y=Y+1 <054>
1330 GOSUB 1500 <096>
1340 RETURN <128>
1400 REM *** CURSOR RAUF *** <036>
1410 GOSUB 1500 <176>
1420 IF Y>0 THEN Y=Y-1 <223>
1430 GOSUB 1500 <196>
1440 RETURN <228>
1500 REM *** CURSOR SETZEN/LOESCHEN *** <015>
1510 IF PEEK(1076+X+Y*40)>128 THEN POKE 10
      76+X+Y*40,PEEK(1076+X+Y*40)AND 127:RE
      TURN <172>
1520 POKE 1076+X+Y*40,PEEK(1076+X+Y*40)OR
      128:RETURN <244>
1600 REM *** PUNKT SETZEN/LOESCHEN *** <234>
1610 IF PEEK(1076+X+Y*40)=160 THEN POKE 10
      76+X+Y*40,209:RETURN <021>
1620 IF PEEK(1076+X+Y*40)=209 THEN POKE 10
      76+X+Y*40,160:RETURN <186>
1700 REM *** NEUES ZEICHEN *** <070>
1710 GOSUB 1500:INPUT "{HOME,21DOWN}BREITE {
      3SPACE}";B(L) <246>
1720 V=0:W=0:GOSUB 3000 <178>
1730 V=8:W=1:GOSUB 3000 <061>
1740 V=320:W=16:GOSUB 3000 <097>
1750 V=328:W=17:GOSUB 3000 <117>
1760 RETURN <038>
2000 REM *** TEILZEICHEN SETZEN *** <077>
2010 FOR K=0 TO 7 <061>
2020 FOR M=0 TO 7 <087>
2030 IF (B(L*32+3+K*2+W)AND 2↑M)=0 THEN 205
      0 <124>
2040 POKE 1076+K*40+(7-M)+V,81 <014>
2050 POKE 55348+K*40+(7-M)+V,6 <064>
2060 NEXT M,K <187>
2070 RETURN <096>
3000 REM *** TEILZEICHEN HOLEN *** <130>
3010 FOR K=0 TO 7:B(L*32+3+K*2+W)=0 <011>
3020 FOR M=0 TO 7 <071>
3030 IF PEEK(1076+K*40+(7-M)+V)=81 THEN B(
      L*32+3+K*2+W)=B(L*32+3+K*2+W)OR 2↑M <102>
3040 NEXT M,K <149>
3050 RETURN <058>
5000 REM *** BILDSCHIRMAUFBAU *** <023>
5010 PRINT "{CLR,11SPACE,RVSON,18SPACE}" <050>
5020 FOR K=1 TO 16 <106>
5030 PRINT "{11SPACE,RVSON,SPACE,16RIGHT,SP
      ACE}" <160>
5040 NEXT <224>
5050 PRINT "{11SPACE,RVSON,18SPACE}" <188>
5060 RETURN <036>

```

Listing 2. Änderungsprogramm für die drei großen Zeichensätze